NAG Library Routine Document

G04AGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G04AGF performs an analysis of variance for a two-way hierarchical classification with subgroups of possibly unequal size, and also computes the treatment group and subgroup means. A fixed effects model is assumed.

2 Specification

```
 SUBROUTINE G04AGF(Y, N, K, LSUB, NOBS, L, NGP, GBAR, SGBAR, GM, SS, IDF,
1                  F, FP, IFAIL)
 INTEGER           N, K, LSUB(K), NOBS(L), L, NGP(K), IDF(4), IFAIL
 double precision  Y(N), GBAR(K), SGBAR(L), GM, SS(4), F(2), FP(2)
```

3 Description

In a two-way hierarchical classification, there are k (≥ 2) treatment groups, the ith of which is subdivided into l_i treatment subgroups. The jth subgroup of group i contains n_{ij} observations, which may be denoted by

$$y_{1ij}, y_{2ij}, \dots, y_{n_{ij}ij}.$$

The general observation is denoted by y_{mij}, being the mth observation in subgroup j of group i, for $1 \leq i \leq k$, $1 \leq j \leq l_i$, $1 \leq m \leq n_{ij}$.

The following quantities are computed

(i) The subgroup means

$$\bar{y}_{.ij} = \frac{\sum_{m=1}^{n_{ij}} y_{\mathrm{mij}}}{n_{ij}}$$

(ii) The group means

$$\bar{y}_{.i.} = \frac{\sum_{j=1}^{l_i} \sum_{m=1}^{n_{ij}} y_{\mathrm{mij}}}{\sum_{j=1}^{l_i} n_{ij}}$$

(iii) The grand mean

$$\bar{y}_{...} = \frac{\sum_{i=1}^{k} \sum_{j=1}^{l_i} \sum_{m=1}^{n_{ij}} y_{\mathrm{mij}}}{\sum_{i=1}^{k} \sum_{j=1}^{l_i} n_{ij}}$$

(iv) The number of observations in each group

$$n_{i.} = \sum_{j=1}^{l_i} n_{ij}$$

(v) Sums of squares

$$\text{Between groups} \quad = \mathrm{SS}_g = \sum_{i=1}^{k} n_{i.}\left(\bar{y}_{.i.} - \bar{y}_{...}\right)^2$$

$$\text{Between subgroups within groups} \quad = \mathrm{SS}_{sg} = \sum_{i=1}^{k}\sum_{j=1}^{l_i} n_{ij}\left(y_{.ij} - \bar{y}_{.i.}\right)^2$$

$$\text{Residual (within subgroups)} \quad = \mathrm{SS}_{\mathrm{res}} = \sum_{i=1}^{k}\sum_{j=1}^{l_i}\sum_{m=1}^{n_{ij}}\left(y_{\mathrm{mij}} - \bar{y}_{.ij}\right)^2 = \mathrm{SS}_{\mathrm{tot}} - \mathrm{SS}_g - \mathrm{SS}_{sg}$$

$$\text{Corrected total} \quad = \mathrm{SS}_{\mathrm{tot}} = \sum_{i=1}^{k}\sum_{j=1}^{l_i}\sum_{m=1}^{n_{ij}}\left(y_{\mathrm{mij}} - \bar{y}_{...}\right)^2$$

(vi) Degrees of freedom of variance components

Between groups:	$k - 1$
Subgroups within groups:	$l - k$
Residual:	$n - l$
Total:	$n - 1$

where

$$l = \sum_{i=1}^{k} l_i,$$

$$n = \sum_{i=1}^{k} n_{i.}$$

(vii) F ratios. These are the ratios of the group and subgroup mean squares to the residual mean square.

$$\text{Groups} \quad F_1 = \frac{\text{Between groups sum of squares}/(k-1)}{\text{Residual sum of squares}/(n-l)} = \frac{\mathrm{SS}_g/(k-1)}{\mathrm{SS}_{\mathrm{res}}/(n-l)}$$

$$\text{Subgroups} \quad F_2 = \frac{\text{Between subgroups (within group) sum of squares}/(l-k)}{\text{Residual sum of squares}/(n-l)} = \frac{\mathrm{SS}_{sg}/(l-k)}{\mathrm{SS}_{\mathrm{res}}/(n-l)}$$

If either F ratio exceeds 9999.0, the value 9999.0 is assigned instead.

(viii) F significances. The probability of obtaining a value from the appropriate F-distribution which exceeds the computed mean square ratio.

Groups $\quad p_1 = \mathrm{Prob}\left(F_{(k-1),(n-l)} > F_1\right)$

Subgroups $\quad p_2 = \mathrm{Prob}\left(F_{(l-k),(n-l)} > F_2\right)$

where F_{ν_1,ν_2} denotes the central F-distribution with degrees of freedom ν_1 and ν_2.

If any $F_i = 9999.0$, then p_i is set to zero, $i = 1, 2$.

4 References

Kendall M G and Stuart A (1976) *The Advanced Theory of Statistics (Volume 3)* (3rd Edition) Griffin

Moore P G, Shirley E A and Edwards D E (1972) *Standard Statistical Calculations* Pitman

5 Parameters

1: Y(N) – ***double precision*** array *Input*

On entry: the elements of Y must contain the observations y_{mij} in the following order:

$$y_{111}, y_{211}, \ldots, y_{n_{11}11}, y_{112}, y_{212}, \ldots, y_{n_{12}12}, \ldots, y_{11l_1}, \ldots,$$

$$y_{n_{1l_1}1l_1}, \ldots, y_{1ij}, \ldots, y_{n_{ij}ij}, \ldots, y_{1kl_k}, \ldots, y_{n_{kl_k}kl_k}.$$

In words, the ordering is by group, and within each group is by subgroup, the members of each subgroup being in consecutive locations in Y.

2: N – INTEGER *Input*

On entry: n, the total number of observations.

3: K – INTEGER *Input*

On entry: k, the number of groups.

Constraint: $\mathrm{K} \geq 2$.

4: LSUB(K) – INTEGER array *Input*

On entry: the number of subgroups within group i, l_i, for $i = 1, 2, \ldots, k$.

Constraint: $\mathrm{LSUB}(i) > 0$, for $i = 1, 2, \ldots, k$.

5: NOBS(L) – INTEGER array *Input*

On entry: the numbers of observations in each subgroup, n_{ij}, in the following order:

$$n_{11}, n_{12}, \ldots, n_{1l_1}, n_{21}, \ldots, n_{2l_2}, \ldots, n_{k1}, \ldots, n_{kl_k}$$

Constraint: $n = \sum_{i=1}^{k} \sum_{j=1}^{l_i} n_{ij}$, that is $\mathrm{N} = \sum_{i=1}^{l} \mathrm{NOBS}(i)$ and $\mathrm{NOBS}(i) > 0$, for $i = 1, 2, \ldots, l$.

6: L – INTEGER *Input*

On entry: l, the total number of subgroups.

Constraint: $\mathrm{L} = \sum_{i=1}^{k} \mathrm{LSUB}(i)$.

7: NGP(K) – INTEGER array *Output*

On exit: the total number of observations in group i, $n_{i.}$, for $i = 1, 2, \ldots, k$.

8: GBAR(K) – ***double precision*** array *Output*

On exit: the mean for group i, $\bar{y}_{.i.}$, for $i = 1, 2, \ldots, k$.

9: SGBAR(L) – ***double precision*** array *Output*

On exit: the subgroup means, $\bar{y}_{.ij}$, in the following order:

$$\bar{y}_{.11}, \bar{y}_{.12}, \ldots, \bar{y}_{.1l_1}, \bar{y}_{.21}, \bar{y}_{.22}, \ldots, \bar{y}_{.2l_2}, \ldots, \bar{y}_{.k1}, \bar{y}_{.k2}, \ldots, \bar{y}_{.kl_k}.$$

10: GM – ***double precision*** *Output*

On exit: the grand mean, $\bar{y}_{...}$.

11: SS(4) – ***double precision*** array *Output*

On exit: contains the sums of squares for the analysis of variance, as follows;

SS(1) = Between group sum of squares, SS_g,

SS(2) = Between subgroup within groups sum of squares, SS_{sg},

SS(3) = Residual sum of squares, SS_{res},

SS(4) = Corrected total sum of squares, SS_{tot}.

12: IDF(4) – INTEGER array *Output*

On exit: contains the degrees of freedom attributable to each sum of squares in the analysis of variance, as follows:

IDF(1) = Degrees of freedom for between group sum of squares,

IDF(2) = Degrees of freedom for between subgroup within groups sum of squares,

IDF(3) = Degrees of freedom for residual sum of squares,

IDF(4) = Degrees of freedom for corrected total sum of squares.

13: F(2) – ***double precision*** array *Output*

On exit: contains the mean square ratios, F_1 and F_2, for the between groups variation, and the between subgroups within groups variation, with respect to the residual, respectively.

14: FP(2) – ***double precision*** array *Output*

On exit: contains the significances of the mean square ratios, p_1 and p_2 respectively.

15: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, $\mathrm{K} \leq 1$.

IFAIL $= 2$

On entry, $\mathrm{LSUB}(i) \leq 0$, for some $i = 1, 2, \ldots, k$.

IFAIL $= 3$

On entry, $\mathrm{L} \neq \sum_{i=1}^{k} \mathrm{LSUB}(i)$

IFAIL $= 4$

On entry, $\mathrm{NOBS}(i) \leq 0$, for some $i = 1, 2, \ldots, l$.

IFAIL = 5

On entry, $\mathrm{N} \neq \sum_{i=1}^{l} \mathrm{NOBS}(i)$.

IFAIL = 6

The total corrected sum of squares is zero, indicating that all the data values are equal. The means returned are therefore all equal, and the sums of squares are zero. No assignments are made to IDF, F, and FP.

IFAIL = 7

The residual sum of squares is zero. This arises when either each subgroup contains exactly one observation, or the observations within each subgroup are equal. The means, sums of squares, and degrees of freedom are computed, but no assignments are made to F and FP.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

The time taken by G04AGF increases approximately linearly with the total number of observations, n.

9 Example

This example has two groups, the first of which consists of five subgroups, and the second of three subgroups. The numbers of observations in each subgroup are not equal. The data represent the percentage stretch in the length of samples of sack kraft drawn from consignments (subgroups) received over two years (groups). For details see Moore *et al.* (1972).

9.1 Program Text

```
*     G04AGF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          K, LMAX, NMAX
      PARAMETER        (K=2,LMAX=8,NMAX=28)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION GM
      INTEGER          I, IFAIL, II, J, L, LI, N, NHI, NIJ, NLO, NSUB
*     .. Local Arrays ..
      DOUBLE PRECISION F(2), FP(2), GBAR(K), SGBAR(LMAX), SS(4), Y(NMAX)
      INTEGER          IDF(4), LSUB(K), NGP(K), NOBS(LMAX)
*     .. External Subroutines ..
      EXTERNAL         G04AGF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G04AGF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Data values'
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Group  Subgroup  Observations'
      LSUB(1) = 5
      LSUB(2) = 3
      L = LSUB(1) + LSUB(2)
      IF (L.LE.LMAX) THEN
         READ (NIN,*) (NOBS(I),I=1,L)
         N = 0
         DO 20 I = 1, L
```

```
            N = N + NOBS(I)
   20    CONTINUE
         IF (N.LE.NMAX) THEN
            READ (NIN,*) (Y(I),I=1,N)
            IFAIL = 1
            NSUB = 0
            NLO = 1
            DO 60 I = 1, K
               LI = LSUB(I)
               DO 40 J = 1, LI
                  NSUB = NSUB + 1
                  NIJ = NOBS(NSUB)
                  NHI = NLO + NIJ - 1
                  WRITE (NOUT,99999) I, J, (Y(II),II=NLO,NHI)
                  NLO = NLO + NIJ
   40          CONTINUE
   60       CONTINUE
*
            CALL G04AGF(Y,N,K,LSUB,NOBS,L,NGP,GBAR,SGBAR,GM,SS,IDF,F,FP,
     +                  IFAIL)
*
            IF (IFAIL.NE.0) THEN
               WRITE (NOUT,*)
               WRITE (NOUT,99997) 'Failed in G04AGF. IFAIL = ', IFAIL
            ELSE
               WRITE (NOUT,*)
               WRITE (NOUT,*) 'Subgroup means'
               WRITE (NOUT,*)
               WRITE (NOUT,*) '   Group  Subgroup  Mean'
               II = 0
               DO 100 I = 1, K
                  LI = LSUB(I)
                  DO 80 J = 1, LI
                     II = II + 1
                     WRITE (NOUT,99998) I, J, SGBAR(II)
   80             CONTINUE
  100          CONTINUE
               WRITE (NOUT,*)
               WRITE (NOUT,99996) '    Group 1 mean = ', GBAR(1),
     +           '   (', NGP(1), ' observations)'
               WRITE (NOUT,99996) '    Group 2 mean = ', GBAR(2),
     +           '   (', NGP(2), ' observations)'
               WRITE (NOUT,99996) '    Grand mean   = ', GM, '   (', N,
     +           ' observations)'
               WRITE (NOUT,*)
               WRITE (NOUT,*) 'Analysis of variance table'
               WRITE (NOUT,*)
               WRITE (NOUT,*)
     +           '   Source                SS     DF  F ratio  Sig'
               WRITE (NOUT,*)
               WRITE (NOUT,99995) 'Between groups         ', SS(1),
     +           IDF(1), F(1), FP(1)
               WRITE (NOUT,99995) 'Bet sbgps within gps   ', SS(2),
     +           IDF(2), F(2), FP(2)
               WRITE (NOUT,99995) 'Residual               ', SS(3),
     +           IDF(3)
               WRITE (NOUT,*)
               WRITE (NOUT,99995) 'Total                  ', SS(4),
     +           IDF(4)
            END IF
         END IF
      END IF
*
99999 FORMAT (1X,I5,I9,4X,10F4.1)
99998 FORMAT (1X,I6,I8,F10.2)
99997 FORMAT (1X,A,I5)
99996 FORMAT (1X,A,F4.2,A,I2,A)
99995 FORMAT (1X,A,F5.3,I5,F7.2,F8.3)
      END
```

9.2 Program Data

```
G04AGF Example Program Data
 5 3 3 3 2 3 5 3
 2.1 2.4 2.0 2.0 2.0 2.4 2.1 2.2 2.4 2.2
 2.6 2.4 2.4 2.5 1.9 1.7 2.1 1.5 2.0 1.9
 1.7 1.9 1.9 1.9 2.0 2.1 2.3
```

9.3 Program Results

```
 G04AGF Example Program Results

 Data values

  Group  Subgroup  Observations
     1       1      2.1 2.4 2.0 2.0 2.0
     1       2      2.4 2.1 2.2
     1       3      2.4 2.2 2.6
     1       4      2.4 2.4 2.5
     1       5      1.9 1.7
     2       1      2.1 1.5 2.0
     2       2      1.9 1.7 1.9 1.9 1.9
     2       3      2.0 2.1 2.3

 Subgroup means

    Group  Subgroup  Mean
      1       1      2.10
      1       2      2.23
      1       3      2.40
      1       4      2.43
      1       5      1.80
      2       1      1.87
      2       2      1.86
      2       3      2.13

     Group 1 mean = 2.21   (16 observations)
     Group 2 mean = 1.94   (11 observations)
     Grand mean   = 2.10   (27 observations)

 Analysis of variance table

    Source                 SS      DF  F ratio  Sig

 Between groups         0.475    1  16.15   0.001
 Bet sbgps within gps   0.816    6   4.63   0.005
 Residual               0.559   19

 Total                  1.850   26
```